# EXST 7015
# Fall 2014
# Lab 10: ANOVA and Post ANOVA Test

**OBJECTIVES**

Analysis of variance (ANOVA) is the most commonly used technique for comparing the means of groups of measurement data. There are lots of different experimental designs that can be analyzed with different kinds of ANOVA. For this week's lab, the most basic type of ANOVA, one-way ANOVA will be introduced. In a one-way ANOVA, there is one continuous variable (Response Variable) and one categorical variable (Treatment). Multiple observations of the response variable are made for each level of the treatment.

One-way ANOVA corresponds to the completely randomized designed experiment (CRD) with one fixed treatment effect, with its linear model as the following:

$$\mathbf{Y_{ij} = \mu + \tau_i + \varepsilon_{ij} = \mu_i + \varepsilon_{ij} \ \ ( i = 1, 2, \ldots, t; j = 1, 2, \ldots, n)}$$

Where $\boldsymbol{\mu}$ is the overall mean; $\tau_i$ are the treatment level effects, and $\varepsilon_{ij}$ is the random error. $\boldsymbol{\mu_i}$ is the mean of the $i^{th}$ level of treatment. The null hypothesis test of one-way ANOVA is that the means of the response variable are the same for the different levels of treatment ($H_0$: $\boldsymbol{\mu_1 = \mu_2 = \ldots = \mu_t}$); the alternative hypothesis is that they are not all the same.

There are three assumptions need to be considered for ANOVA: the treatments are independently sampled; residuals or deviation of observations within groups should be normally distributed (evaluated by residual plot, normality test); and the variance from each level of treatment is the same (i.e. homogenous variance, Bartlett test is preferred method to evaluate the homogeneity of variance).

ANOVA usually proceeds with an F-test of the MSTreatment (d.f.= t-1) over MSError (d.f. = t(n-1)). The MSError estimates a variance $\sigma^2_\varepsilon$, and MSTreatment estimates the same $\sigma^2_\varepsilon$ plus the difference between the levels of treatment ($\sigma^2_\varepsilon + n\sigma^2_\tau$). So the F-test can be written as the following:

$$F = MSTreatment/MSError = (\sigma^2_\varepsilon + n\sigma^2_\tau) / \sigma^2_\varepsilon$$

It is One Tailed F-test since the variance of treatment is expected to be large if the null hypothesis is rejected.

Once the null hypothesis of ANOVA is rejected, i.e., the significant difference among the levels of treatment is concluded from F-Test of ANOVA, Post-ANOVA tests are needed to determine how the treatment levels are interrelated. Common Post hoc tests will be introduced in this lab.

ANOVA and post-ANOVA tests can be performed by suing PROC GLM that has been used in precious labs to fit the regression models. Actually its form does not look very different from what we have used for the regression model, with only a couple of additional statements and options that will be explained following the SAS codes. In addition, **mixed model analysis** will be briefly introduced to construct ANOVA and post-ANOVA tests by using **PROC MIXED,** which has many options not available in the traditional analysis of variance.

**LABORATORY INSTRUCTIONS**

**Housekeeping Statements**

```
dm 'log; clear; output; clear';
options nodate nocenter pageno = 1 ls=78 ps=53;
title1 'EXST7015 lab 10, Name, Section#';
ods rtf  file = ‘c:/temp/lab9.rtf’;
ods html file = ‘c:/temp/lab9.html’;
```

**Data set**

The word lists are standard audiology tools for assessing hearing. They are calibrated to be equally difficult to perceive. However, the original calibration was performed with normal hearing subjects and no noise background. The experimenter wished to determine whether the lists were still equally difficult to understand in the presence of a noisy background. 24 subjects with normal hearing listened to standard audiology tapes of four standard 50-English-word lists at low volume with a noisy background. They repeated the words and were scored correct or incorrect in their perception of the words. The percent of words heard correctly was recorded and will be used as the response variable. Note that this is actually a randomized block design, which has not been covered. Therefore, we will pretend that the data was collected without control on the subjects and treat it as a CRD. Detailed information and references can be found at http://lib.stat.cmu.edu/DASL/Datafiles/Hearing.html. The variables in the dataset are:

1. List: ID of the word lists, which has four levels (Treatment):
   - List 1
   - List 2
   - List 3
   - List 4
2. Score: Percent of the words heard correctly (Response Variable).

```
data hearing;
input id list $ score;
drop id;
cards;
1 List1 28
2 List1 24
3 List1 32
4 List1 30
5 List1 34
6 List1 30
7 List1 36
8 List1 32
9 List1 48
10 List1 32
11 List1 32
12 List1 38
13 List1 32
14 List1 40
15 List1 28
16 List1 48
17 List1 34
```

18 List1 28
19 List1 40
20 List1 18
21 List1 20
22 List1 26
23 List1 36
24 List1 40
1 List2 20
2 List2 16
3 List2 38
4 List2 20
5 List2 34
6 List2 30
7 List2 30
8 List2 28
9 List2 42
10 List2 36
11 List2 32
12 List2 36
13 List2 28
14 List2 38
15 List2 36
16 List2 28
17 List2 34
18 List2 16
19 List2 34
20 List2 22
21 List2 20
22 List2 30
23 List2 20
24 List2 44
1 List3 24
2 List3 32
3 List3 20
4 List3 14
5 List3 32
6 List3 22
7 List3 20
8 List3 26
9 List3 26
10 List3 38
11 List3 30
12 List3 16
13 List3 36
14 List3 32
15 List3 38
16 List3 14
17 List3 26
18 List3 14
19 List3 38
20 List3 20
21 List3 14
22 List3 18
23 List3 22
24 List3 34
1 List4 26
2 List4 24

```
3 List4 22
4 List4 18
5 List4 24
6 List4 30
7 List4 22
8 List4 28
9 List4 30
10 List4 16
11 List4 18
12 List4 34
13 List4 32
14 List4 34
15 List4 32
16 List4 18
17 List4 20
18 List4 20
19 List4 40
20 List4 26
21 List4 14
22 List4 14
23 List4 30
24 List4 42
;

proc boxplot data=hearing;
plot score*list;
run;
```

**PROC BOXPLOT** will constructs boxplots for each level of treatment.

## Analysis of Variance using PROC GLM

```
proc glm data=hearing;
  class list;
  model score = list ;
  means list / lsd tukey hovtest=bartlett;
  output out=outdata p=pred residual=resid;
run;
proc univariate data=outdata normal plot;
  var resid;
run;
proc plot data=outdata;
  plot resid*pred;
run;
```

**CLASS statement** tells the GLM procedure to treat the variable as categorical variable.

**MEANS statement:** tests for differences between different pairs of treatment levels.

**LSD TUKEY** Options following the **MEANS statement** specifies post hoc tests. Other methods such as Scheffé and Dunnett's can be specified as needed.

**HOVTEST = Bartlett:** Bartlett's test is generally considered the best test for homogeneity of variance. This test uses a calculation which follows a Chi-Square distribution. The null hypothesis test is that the model output residuals are distributed homogeneously about the hypothesized mean of zero. It must be used with the MEANS statement in the PROC GLM.

In addition to Bartlett, other methods such as **Levene** and **Obrien** could be used to test the homogeneity of variance.

## Mixed Model Analysis using PROC MIXED

```
proc mixed data=hearing;
  class list;
  model score=list / ddfm=satterth outp=outdata;
  repeated / group=list;
  lsmeans list / adjust=tukey pdiff;
run;
proc univariate data=outdata normal plot;
var resid;
run;
ods rtf close;
```

**REPEATED /GROUP=List:** returns the results of the homogeneity of variance.

**LSMEAN effect /adjust = <adjust method> pdiff:** The **LSMEANS statement** tests for differences between different pairs of treatment levels, where **EFFECT** is the treatment effect variable, and the **option adjust=** specifies adjustment methods. If this options Statement is omitted then SAS will conduct **LSD** tests by default. **Pdiff** will provide actual probabilities for each pairwise comparison.

**LAB ASSIGNMENT**

1. Write the linear model to test the hypothesis that there is no treatment effect. Clearly describe each term in the model, and the range of the subscripts. Write the null hypothesis that you are testing.

2. Use both proc **glm** and proc **mixed** to test the hypothesis you stated in Question 1. Report your results, including your F-value, p-value, and conclusions.

3. List the assumptions necessary for your analysis and determine whether they have been violated. Include any relevant SAS output in you report.

4. If there is a significant treatment effect, describe which pairs of means are different. Explain which adjustment method you chose.